unittest
{
	assert( [ "2", "ab", "ca" ].parse.expand.solve == 2 );
	assert( [ "4", "aaaa", "aaaa", "aaaa", "aaaa" ].parse.expand.solve == 16 );
	assert( [ "5", "abcde", "fghij", "klmno", "pqrst", "uvwxy" ].parse.expand.solve == 0 );
}



import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto n = input.front.to!( long );
	auto ss = new dchar[][]( n, n );
	foreach( ref s; ss )
	{
		input.popFront;
		s = input.front.to!( dchar[] );
	}
	return tuple( n, ss );
}

auto solve( long n, dchar[][] ss )
{
	auto ans = 0L;
	foreach( a; 0L .. ss.length )
	{
		// シンメトリー判定
		auto c = true;
		LOOP: for( auto i = 0L; i < ss.length; i++ )
		{
			for( auto j = 0L; j < ss[ i ].length; j++ )
			{
				c = ( ss[ i ][ j ] == ss[ j ][ i ] );
				if( !c ) break LOOP;
			}
		}
		
		// a,bのときとa+k,b+kのときの判定結果は同じになるので、n通りとして加算
		if( c ) ans += n;
		
		// 盤面をずらす
		ss = ss[ 1 .. $ ] ~ ss[ 0 ];
	}
	return ans;
}
